Collections in Application Architecture and Design
/* Generic global types */

CREATE OR REPLACE TYPE tnt_num IS TABLE OF NUMBER;

CREATE OR REPLACE TYPE tnt_str IS TABLE OF VARCHAR2(255);

CREATE OR REPLACE TYPE tnt_dtm IS TABLE OF DATE;

/* Generic common associative array types and their empties */

CREATE OR REPLACE PACKAGE t
/******************************************************************************
* $Revision:   1.2  $  $Author:   bcoulam  $
* $Modtime:   Jan 03 2003 13:48:32  $
*
* Description:
*  Collection of application specific PL/SQL table types and user-defined
*  types (subtypes, records and tables of records).
*
Usually tied to specific tables.
******************************************************************************/
IS 

-------------------------------------------------------------------------------
-- subtypes
SUBTYPE t_maxvc2col IS VARCHAR2(4000);
SUBTYPE t_maxvc2 IS VARCHAR2(32767);
SUBTYPE t_msg IS t_maxvc2col;
SUBTYPE t_rc IS PLS_INTEGER;
SUBTYPE t_return_cd IS PLS_INTEGER;

-- THESE ARE THE NEW PL/SQL TABLES. USE THESE.
TYPE tab IS TABLE OF BOOLEAN INDEX BY BINARY_INTEGER;
TYPE tad IS TABLE OF DATE INDEX BY BINARY_INTEGER;
TYPE tan IS TABLE OF NUMBER
INDEX BY BINARY_INTEGER;
TYPE tas1 IS TABLE OF VARCHAR2(1) INDEX BY BINARY_INTEGER;
TYPE tas5 IS TABLE OF VARCHAR2(5) INDEX BY BINARY_INTEGER;
TYPE tas30 IS TABLE OF VARCHAR2(30) INDEX BY BINARY_INTEGER;
TYPE tas60 IS TABLE OF VARCHAR2(60) INDEX BY BINARY_INTEGER;
TYPE tas255 IS TABLE OF VARCHAR2(255) INDEX BY BINARY_INTEGER;
TYPE tas2000 IS TABLE OF VARCHAR2(2000) INDEX BY BINARY_INTEGER;
TYPE tasMaxCol IS TABLE OF t_maxvc2col INDEX BY BINARY_INTEGER;
TYPE tasMax IS TABLE OF t_maxvc2 INDEX BY BINARY_INTEGER;

-------------------------------------------------------------------------------
-- Empty arrays of the generic types defined above
-- These are to be used in 9iAS PL/SQL-web parameter lists as default values
gab_empty
tab;
gad_empty
tad;
gan_empty
tan;
gas1_empty
tas1;
gas5_empty    tas5;
gas30_empty    tas30;
gas60_empty    tas60;
gas255_empty  tas255;
gas2000_empty    tas2000;
gas4000_empty    t_maxvc2col;
gasMaxCol_empty tasMaxCol;
gasMax_empty    tasMax;

END t;

/

CREATE OR REPLACE PACKAGE  strings
/******************************************************************************
* $Revision:   1.4  $  $Author:   bcoulam  $
* $Modtime:   Feb 19 2003 17:44:50  $
*
* Description:
*  Package bundling generic routines for string manipulation/handling.  
*
******************************************************************************/
   AUTHID CURRENT_USER
IS 
-- Public constants, types, cursors, procedures, functions, etc.

-------------------------------------------------------------------------------
-- generic print control constants
TAB   CONSTANT VARCHAR2(1) := CHR(9); -- A single tab character
CR    CONSTANT VARCHAR2(1) := CHR(13); -- A single carriage return character (^M)
LF    CONSTANT VARCHAR2(1) := CHR(10); -- A single linefeed character
CRLF  CONSTANT VARCHAR2(2) := CHR(13)||CHR(10);
FF    CONSTANT VARCHAR2(1) := CHR(12); -- A form feed
SP    CONSTANT VARCHAR2(1) := CHR(32); -- A space
DEL   CONSTANT VARCHAR2(1) := CHR(127); -- old delete, non-printing

-- empty collections
gts tnt_str := tnt_str();

-------------------------------------------------------------------------------
-- This function takes in a string and a page (width) to "break" the
-- string into lines of manageable width,. separated by linefeeds.
-------------------------------------------------------------------------------
FUNCTION fs_break_string
(
   is_str   IN VARCHAR2, -- long string to break up
   in_page   IN NUMBER DEFAULT 50 -- page width in chars
)  RETURN   VARCHAR2;

-------------------------------------------------------------------------------
/*
 When HTML tables are overkill and you've got a result set you want to 
 display in columns, use this function to pad or trim a string down to 
 the desired column width. CAUTION: This will NOT work if the output
 isn't viewed in a fixed-width font, like Courier. A corresponding HTML
 version of this function is in ia_ui package.
*/
-------------------------------------------------------------------------------
FUNCTION fs_ewc
(
   is_s  IN VARCHAR2 DEFAULT NULL, -- string to fill or trim
   in_colsize IN  PLS_INTEGER DEFAULT 1 -- column size desired
)  RETURN VARCHAR2;

-------------------------------------------------------------------------------
/*
This function takes in a string and cleans off:
   ALL non-printing characters from the left and right side
*/
-------------------------------------------------------------------------------
FUNCTION fs_clean_str
(
   is_str   IN VARCHAR2
)  RETURN   VARCHAR2;

-------------------------------------------------------------------------------
/*
Takes a string that is supposed to be a delimited list of strings, e.g.
      1,2,3, 4, 28237
and converts it into a collection of strings.
Obviously, if one of the strings includes a character which is the requested
delimiter, that string will be broken up where it ought not.
*/
-------------------------------------------------------------------------------
FUNCTION fnt_parse_list
(
   is_list IN VARCHAR2
  ,is_delimiter IN VARCHAR2 DEFAULT ','
) RETURN tnt_str;

-------------------------------------------------------------------------------
/*
Takes an array of strings and returns a delimited list of the same in a single
string. The returned string can be up to 32757 bytes long, so the receiving
variable or column should be the same size, or the result should be truncated
using SUBSTR.

Obviously, if one of the strings includes a character which is the requested
delimiter, that string will be broken up where it ought not.
*/
-------------------------------------------------------------------------------
FUNCTION fs_make_list
(
   int_list IN tnt_str
  ,is_delimiter IN VARCHAR2 DEFAULT ','
) RETURN VARCHAR2;

END strings;

/

CREATE OR REPLACE PACKAGE BODY strings
/******************************************************************************
* $Revision:   1.4  $  $Author:   bcoulam  $
* $Modtime:   Feb 19 2003 17:50:22  $
*
******************************************************************************/
IS 

/* package global variables, cursors, etc. ----------------------------------*/

/* private procedures and functions -----------------------------------------*/

/* public procedures and functions ------------------------------------------*/
-------------------------------------------------------------------------------
FUNCTION fs_break_string
(
   is_str IN VARCHAR2,
   in_page IN NUMBER DEFAULT 50
)  RETURN VARCHAR2
IS
   ls_str                        VARCHAR2(10240) := is_str;
   ln_len                        NUMBER := LENGTH (ls_str);
   pos                           NUMBER := 1;
   last_ws                       NUMBER := 0;
   nl_pos                        NUMBER := 0;
BEGIN
   IF  (is_str IS NULL) THEN
      RETURN NULL;
   END IF;
   
   LOOP
      LOOP
         nl_pos := INSTR (ls_str, CHR(10), pos);
         IF ( (NL_POS <  (in_page + pos - 1))
         AND (nl_pos > 0)) THEN
            pos := nl_pos + 1;
         ELSE
            EXIT;
         END IF;
      END LOOP;
      
      -- if a line break occurs past a page length...
      IF  ( (INSTR (ls_str, CHR (10), pos) >  (in_page + pos - 1))
      OR  (LENGTH (ls_str) >  (in_page))) THEN
         
         -- get the last white space of that page
         last_ws :=
            GREATEST (INSTR (ls_str, ' ',  (pos + in_page - ln_len - 1)),
            INSTR (ls_str, CHR (9),  (pos + in_page - ln_len - 1))
            );
         
         -- if there was no white space in that page...
         IF  (last_ws < pos) THEN
            -- break word
            ls_str :=
               SUBSTR (ls_str, 1, pos + in_page - 1) ||
               CHR (10) ||
               SUBSTR (ls_str, pos + in_page);
            ln_len := ln_len + 1;
            pos := pos + in_page + 1;

         ELSE
            -- convert white space to return
            ls_str :=
               SUBSTR (ls_str, 1, last_ws - 1) ||
               CHR (10) ||
               SUBSTR (ls_str, last_ws + 1);
            ln_len := ln_len + 1;
            pos := last_ws + 1;
         END IF;
      ELSE
         pos := pos + in_page;
      END IF;
      
      EXIT WHEN  (LENGTH (ls_str) - in_page < pos);
   END LOOP;
   RETURN  (ls_str);
END fs_break_string;

-------------------------------------------------------------------------------
FUNCTION fs_ewc
(
   is_s  IN VARCHAR2 DEFAULT NULL,
   in_colsize IN  PLS_INTEGER DEFAULT 1
)  RETURN VARCHAR2
IS
   s VARCHAR2 (2000) := NVL(is_s,'');
   slen PLS_INTEGER := NVL(length(is_s),0);
   diff  PLS_INTEGER := 0;
BEGIN
   IF (slen < in_colsize) THEN
      diff := in_colsize - slen;
      
      FOR i IN 1..diff LOOP
         s := s || ' ';
      END LOOP;
   ELSIF (slen > in_colsize) THEN
      s := substr(s,1,in_colsize);
   END IF;
   
   RETURN s;
END fs_ewc;

-------------------------------------------------------------------------------
FUNCTION fs_clean_str
(
   is_str   IN VARCHAR2
)  RETURN   VARCHAR2
IS
   ls_char CHAR(1);
   ln_idx PLS_INTEGER := 1;
   ln_left_pos PLS_INTEGER := 0;
   ln_right_pos PLS_INTEGER := 0;
   
BEGIN
   
   IF (is_str IS NULL) THEN
      RETURN NULL;
   ELSE
      ls_char := SUBSTR(is_str, ln_idx, 1);
   END IF;
   
   <<clean_left>>
   LOOP
      EXIT clean_left WHEN ( -- exits when first printing character is found
                 ls_char > SP -- no spaces or non-printing
                 AND
                 ls_char != DEL -- no delete char
                );
      ln_idx := ln_idx + 1;
      ls_char := SUBSTR(is_str, ln_idx, 1);
   END LOOP;
   
   ln_left_pos := ln_idx; -- store position of first printing character
   
   ln_idx := LENGTH(is_str);
   ls_char := SUBSTR(is_str, ln_idx, 1); --re-init starting at far right
   
   <<clean_right>>
   LOOP
      EXIT clean_right WHEN ( -- exits when first printing character is found
                 ls_char > SP -- no spaces or non-printing
                 AND
                 ls_char != DEL -- no delete char
                );
      ln_idx := ln_idx - 1;
      ls_char := SUBSTR(is_str, ln_idx, 1);
   END LOOP;
   
   ln_right_pos := ln_idx;  -- store position of last printing character

   RETURN SUBSTR(is_str, ln_left_pos, (ln_right_pos - ln_left_pos)+1);
   
END fs_clean_str;

-------------------------------------------------------------------------------
FUNCTION fnt_parse_list
(
   is_list IN VARCHAR2
  ,is_delimiter IN VARCHAR2 DEFAULT ','
) RETURN tnt_str
IS
   lnt_tokens tnt_str := tnt_str();
   ls_remains VARCHAR2(32767) := is_list;
BEGIN
   WHILE (ls_remains IS NOT NULL) LOOP
      lnt_tokens.EXTEND;
      SELECT
         SUBSTR(ls_remains,1,
            DECODE(INSTR(ls_remains,is_delimiter)
            ,0,LENGTH(ls_remains)+1,
            INSTR(ls_remains,is_delimiter)
            )-1)
        ,SUBSTR(ls_remains,
           DECODE(INSTR(ls_remains,is_delimiter)
           ,0,LENGTH(ls_remains),
           INSTR(ls_remains,is_delimiter)
           )+1)
      INTO
         lnt_tokens(lnt_tokens.COUNT)
        ,ls_remains
      FROM dual;
   END LOOP;
   RETURN lnt_tokens;
END fnt_parse_list;

-------------------------------------------------------------------------------
FUNCTION fs_make_list
(
   int_list IN tnt_str
  ,is_delimiter IN VARCHAR2 DEFAULT ','
) RETURN VARCHAR2
IS
   ls_str t.t_maxvc2;
BEGIN
   IF (int_list.COUNT > 0) THEN
      FOR i IN int_list.first..int_list.last LOOP
         BEGIN
            IF (i > 1) THEN
              ls_str := ls_str || is_delimiter || int_list(i);
            ELSE -- only hit on first item
               ls_str := int_list(i);
            END IF;
         EXCEPTION
            WHEN NO_DATA_FOUND THEN --for deleted entries in nested table
               NULL; -- move on to next entry (if any)
         END;
      END LOOP;
   END IF;
   RETURN ls_str;
END fs_make_list;

END strings;
/

CREATE OR REPLACE PACKAGE numbers
/******************************************************************************
* $Revision:   1.0  $  $Author:   bcoulam  $
* $Modtime:   Feb 19 2003 18:42:32  $
*
* Description:
*  Contains utility routines to deal with numeric types, values and result sets.
*
******************************************************************************/
   AUTHID CURRENT_USER
AS

/* Public types -------------------------------------------------------------*/
/* Public constants, exceptions, etc. ---------------------------------------*/
INTEGER_MASK CONSTANT VARCHAR2(40) := 'FM999G999G999G990D0009';

/* Public cursors -----------------------------------------------------------*/
/* Public functions ---------------------------------------------------------*/
-------------------------------------------------------------------------------
/* Compares two nested tables of number, a permanent type in the COMMON schema.
 It will return a list of any number in set one that doesn't appear in set two.
*/
-------------------------------------------------------------------------------
FUNCTION fnt_get_set_diff
(
   int_tab1 IN tnt_num
,  int_tab2 IN tnt_num
) RETURN tnt_num;

-------------------------------------------------------------------------------
/*
 Takes a string that is supposed to be a delimited list of numbers, e.g.
       1,2,3, 4, 28237
 and converts it into a collection of numbers.
 There is currently no exception handling, so if the caller passes in
 alpha characeters or an inconsistent delimiter, it will bomb hard.
*/
-------------------------------------------------------------------------------
FUNCTION fnt_parse_list
(
   is_list IN VARCHAR2
  ,is_delimiter IN VARCHAR2 DEFAULT ','
) RETURN tnt_num;

-------------------------------------------------------------------------------
/*
Takes an array of numbers and returns a delimited list of the same in a single
string. The returned string can be up to 32757 bytes long, so the receiving
variable or column should be the same size, or the result should be truncated
using SUBSTR.
*/
-------------------------------------------------------------------------------
FUNCTION fs_make_list
(
   int_list IN tnt_num
  ,is_delimiter IN VARCHAR2 DEFAULT ','
) RETURN VARCHAR2;

/* Public procedures --------------------------------------------------------*/


END numbers;
/

CREATE OR REPLACE PACKAGE BODY numbers
/******************************************************************************
* $Revision:   1.0  $  $Author:   bcoulam  $
* $Modtime:   Feb 19 2003 18:41:18  $
*
******************************************************************************/
AS

/* Package global variables, types, exceptions, etc. ------------------------*/
/* Private procedures and functions -----------------------------------------*/
-------------------------------------------------------------------------------
-- Generic cursor for comparing two nested tables of number.
-- Will return any number in set one that doesn't appear in set two.
CURSOR cur_get_set_diff
(
   int_tab1 IN tnt_num
,  int_tab2 IN tnt_num
)
IS
SELECT o.column_value val
FROM   TABLE(CAST(int_tab1 AS tnt_num)) o
MINUS
SELECT n.column_value val
FROM   TABLE(CAST(int_tab2 AS tnt_num)) n;

/* Public procedure and function definitions --------------------------------*/

-------------------------------------------------------------------------------
FUNCTION fnt_get_set_diff
(
   int_tab1 IN tnt_num
,  int_tab2 IN tnt_num
) RETURN tnt_num
IS
   lnt tnt_num := tnt_num();
BEGIN
   FOR lr IN cur_get_set_diff (int_tab1, int_tab2) LOOP
      lnt.EXTEND;
      lnt(lnt.COUNT) := lr.val;
   END LOOP;
   RETURN lnt;
END fnt_get_set_diff;

-------------------------------------------------------------------------------
FUNCTION fnt_parse_list
(
   is_list IN VARCHAR2
  ,is_delimiter IN VARCHAR2 DEFAULT ','
) RETURN tnt_num
IS
   lnt_tokens tnt_num := tnt_num();
   ls_remains VARCHAR2(32767) := is_list;
BEGIN
   WHILE (ls_remains IS NOT NULL) LOOP
      lnt_tokens.EXTEND;
      SELECT
         SUBSTR(ls_remains,1,
            DECODE(INSTR(ls_remains,is_delimiter)
            ,0,LENGTH(ls_remains)+1,
            INSTR(ls_remains,is_delimiter)
            )-1)
        ,SUBSTR(ls_remains,
           DECODE(INSTR(ls_remains,is_delimiter)
           ,0,LENGTH(ls_remains),
           INSTR(ls_remains,is_delimiter)
           )+1)
      INTO
         lnt_tokens(lnt_tokens.COUNT)
        ,ls_remains
      FROM dual;
   END LOOP;
   RETURN lnt_tokens;
END fnt_parse_list;

-------------------------------------------------------------------------------
FUNCTION fs_make_list
(
   int_list IN tnt_num
  ,is_delimiter IN VARCHAR2 DEFAULT ','
) RETURN VARCHAR2
IS
   ls_str t.t_maxvc2;
BEGIN
   IF (int_list.COUNT > 0) THEN
      FOR i IN int_list.first..int_list.last LOOP
         BEGIN
            IF (i > 1) THEN
              ls_str := ls_str || is_delimiter || TO_CHAR(int_list(i));
            ELSE -- only hit on first item
               ls_str := TO_CHAR(int_list(i));
            END IF;
         EXCEPTION
            WHEN NO_DATA_FOUND THEN --for deleted entries in nested table
               NULL; -- move on to next entry (if any)
         END;
      END LOOP;
   END IF;
   RETURN ls_str;
END fs_make_list;

/* Package initializations (rare)  ------------------------------------------*/

END numbers;
/
